

STAT 541

**Using Macros to  
Solve the  
Collation Problem**

■ The term *collate* refers to collecting or arranging (pages) in proper order. SAS provides the convenience of BY-group processing in many of its procedures. BY-group processing produces output for the different groups of interest, but the default output may not be in the desired order.

# Example

```
PROC SORT DATA=MASTER; BY SCHOOL;  
PROC PRINT DATA=MASTER; BY SCHOOL;  
PAGEBY SCHOOL;  
  VAR SSN LAST FIRST MI EXEMPT;  
  TITLE 'Student Names and Exemption Status';  
PROC MEANS DATA=MASTER; BY SCHOOL;  
  VAR GPA;  
  TITLE 'Average GPA';  
PROC FREQ DATA=MASTER; BY SCHOOL;  
  TABLES STANINE;  
  TITLE 'Distribution of Reading Test Stanines';
```

- The preceding example will produce output by school per procedure.
- What if the desired output is to have all the output from each school together across procedures?
- This can be easily solved using a number of different strategies.

# A Solution

```
%MACRO REPORTS (BEGIN, END);  
%DO INDEX=&BEGIN %TO &END;  
OPTIONS PAGENO=1;  
DATA SCHOOL&INDEX;  
  SET MASTER;  
  IF SCHOOL=&INDEX;  
PROC PRINT DATA=SCHOOL&INDEX;  
  BY SCHOOL;  
  VAR SSN LAST FIRST MI EXEMPT;  
  TITLE 'Student Names and Exemption Status';  
PROC MEANS DATA=SCHOOL&INDEX;  
  BY SCHOOL;  
  VAR GPA;  
  TITLE 'Average GPA';  
PROC FREQ DATA=SCHOOL&INDEX;  
  BY SCHOOL;  
  TABLES STANINE;  
  TITLE 'Distribution of Reading Test Stanines';  
%END;  
%MEND REPORTS;  
**THE FOLLOWING STATEMENT GENERATES REPORTS FOR 50 SCHOOLS;  
%REPORTS(1,50)
```

- For the purpose of providing an example, the schools are numbered or indexed conveniently from 1 to 50.
- A data set created for each school is used one at a time with the procedures.
- The macro parameters BEGIN and END are useful in case there is a need to generate reports for just a number of schools.  
%REPORTS(3,3) will generate the reports for just the 3<sup>rd</sup> school.  
%REPORTS(14,25) will generate reports for the 14<sup>th</sup> through 25<sup>th</sup> schools.

# Another Solution

```
%MACRO REPORTS (BEGIN, END);  
%DO INDEX=&BEGIN %TO &END;  
OPTIONS PAGENO=1;  
PROC PRINT DATA=MASTER;  
  BY SCHOOL;  
  VAR SSN LAST FIRST MI EXEMPT;  
  WHERE SCHOOL=&INDEX;  
  TITLE 'Student Names and Exemption Status';  
PROC MEANS DATA=MASTER;  
  BY SCHOOL;  
  VAR GPA;  
  WHERE SCHOOL=&INDEX;  
  TITLE 'Average GPA';  
PROC FREQ DATA=MASTER;  
  BY SCHOOL;  
  TABLES STANINE;  
  WHERE SCHOOL=&INDEX;  
  TITLE 'Distribution of Reading Test Stanines';  
%END;  
%MEND REPORTS;  
  
%REPORTS (1,50)
```

- The WHERE statement was used so that the only data set needed for the procedures is MASTER.